

# Programmierhandbuch

## ProScreener



ProRealTime



# INHALTSVERZEICHNIS

<b>Einführung in den ProScreener</b>	
<b>Kapitel I: Grundkenntnisse</b>	
ProScreener starten.....	2
Benutzung des Fensters 'Top Mover'.....	7
<b>Kapitel II: Die Programmierung des ProScreener</b>	
Anweisung zum Suchen und Filtern.....	8
Volumen-Antizipation.....	11
Multi-Perioden-Suche.....	11
Multi-Titel-Suche.....	13
<b>Kapitel III: Praktische Anwendungen</b>	
Einfache Beispiele.....	16
1) RSI 1 Stunde: Überverkauf.....	16
2) RSI 1 Stunde: Überkauf.....	16
3) Kreuzen von steigenden Gleitenden Durchschnitten.....	16
4) Kreuzen von fallenden Gleitenden Durchschnitten.....	17
Komplexere Beispiele.....	18
1) RSI und steigende Trendwende.....	18
2) RSI und fallende Trendwende.....	19
3) Bullish engulfing.....	20
4) Bear engulfing.....	21
5) Bearish moving average crossing.....	22
6) Bullish moving average crossing.....	23
<b>Index</b>	

## Einführung in den ProScreener

ProScreener ist ein sehr leistungsfähiges Scanningtool, das Ihnen das Durchsuchen ganzer Märkte erlaubt:

- ▶ Das Suchen nach Titeln, die eine oder mehrere von Ihnen festgelegte Bedingungen erfüllen.
- ▶ In einer oder mehreren Zeiteinheiten (z.B. eine Minute und eine Stunde).
- ▶ In Echtzeit oder bei Schlusskurs jedes Barcharts mit einer Tick per Tick-Präzision.

ProScreener integriert die Formelsprache ProBuilder (es ist ratsam, vorher das entsprechende [Handbuch](#) zu lesen) zusammen mit zusätzlichen Erweiterungen, die ausschließlich auf Filterbedingungen anwendbar sind. Die Recherchen in Echtzeit oder hinsichtlich des Schließens des Balkens können auf folgende Zeiteinheiten angewendet werden:

- ▶ 1 Minute
- ▶ 2 Minuten
- ▶ 3 Minuten
- ▶ 5 Minuten
- ▶ 10 Minuten
- ▶ 15 Minuten
- ▶ 30 Minuten
- ▶ 1 Stunde
- ▶ 2 Stunden
- ▶ 3 Stunden
- ▶ 4 Stunden
- ▶ Täglich
- ▶ Wöchentlich

Die ProScreener-Suchergebnisse werden in einer Liste der 30 besten (aus den Ergebnissen ausgewählten) Wertpapiere auf Grundlage des festgelegten Filterkriteriums dargestellt.

Dieses Handbuch versteht sich als Erweiterung des [ProBuilder-Handbuchs](#), kann aber unabhängig von diesem gelesen werden. Dank eines angeleiteten theoretischen Ansatzes und konkreter Beispiele werden Sie sich im Laufe Ihrer Lektüre die Grundlagen der Indikatorenprogrammierung aneignen, die es Ihnen erlauben werden, mithilfe von ProScreener-Befehlen Filter für die verschiedenen auf der Workstation verfügbaren Märkte zu erstellen. Sie finden am Ende dieses Handbuchs einen Index, der Ihnen einen Überblick über die Gesamtheit der in diesem Modul verfügbaren Befehle verschafft.


Wir wünschen Ihnen viel Erfolg und eine interessante Lektüre!



## Kapitel I: Grundkenntnisse

### ProScreener starten

Sie können den ProScreener öffnen, indem Sie zuerst auf "Ansicht" und dann auf "ProScreener" klicken, wie es im unteren Bild angezeigt wird:




Es erscheint dann das ProScreener-Fenster, das es ermöglicht, entweder einen bereits existierenden Code zu wählen oder die Programmierung eines neuen Codes vorzunehmen. Um zum Programmierbereich zu gelangen, klicken Sie auf "ProScreener-Verwaltung". Das Programmierfenster bietet Ihnen dann folgende Wahlmöglichkeiten:

- ▶ Den "Assistenten", der die Festlegung der Suchkriterien ohne Programmierung erlaubt (zu mehr Informationen dieser Funktionalität raten wir Ihnen, sich das Video "[ProScreener: Echtzeitscan mit mehreren Bedingungen ohne einen Programmcode schreiben zu müssen](#)" anzusehen).
- ▶ Die "Programmierung", die das Programmieren komplexerer Codes sowie das Festlegen der Anwendungsparameter ermöglicht.



Das Fenster "Programmierung" ist in 5 Bereiche unterteilt:

1. Programmbereich.
2. Listenauswahl der Märkte.
3. Betrachteter Zeitraum.
4. Sortierung der Ergebnisse.
5. Versenden des Codes und Verwaltung des Fensters.



Im **ersten** Bereich können Sie:


- ▶ Einen **ProScreener** direkt im Textfeld programmieren.
- ▶ Die Hilfsfunktion "Funktionen einfügen" verwenden, die in einem neuen Fenster die zur Verfügung stehenden, in sieben Kategorien aufgeteilten, Funktionen auflistet, um Ihnen bei der Programmierung kontextuell behilflich zu sein.



Beginnen wir die Option "Funktionen einfügen" zu verwenden und klicken hierfür auf die entsprechende Schaltfläche.

Lassen Sie uns als Beispiel die Anweisung "**SCREENER**" benutzen, die die Suche von Bedingungen erlaubt (im Bereich "Befehle ProScreener" verfügbar).

Markieren Sie also das Wort "**SCREENER**" und klicken Sie dann auf "Einfügen": Der Befehl wird in das Programmierfeld eingefügt.



Der Befehl "**SCREENER**" erlaubt es Ihnen, die jeweilige(n) Suchbedingung(en) festzulegen. Die letzte Zeile jedes ProScreener muss mit dem Befehl "**SCREENER**" beginnen.



Nehmen wir an, alle Wertpapiere des SRD Frankreich suchen zu wollen, bei denen der Eröffnungskurs des Balkens (Zeitraum) über dem Schlusskurs des vorherigen Balkens liegt. Wir schreiben in diesem Fall Folgendes:

```
c1 = (Open > Close[1])
```

```
SCREENER[c1]
```

Ist der Code einmal festgelegt, wählen wir im **zweiten** Bereich entweder eine vordefinierte oder selbst erstellte Liste aus, auf die die Suche angewendet werden soll.

*Bemerkung: Die Suche kann nur auf Titel desselben Marktes angewandt werden. Um den ProScreener mit einer selbst definierten Liste zu verwenden, muss darauf geachtet werden, dass alle dort vorhandenen Titel demselben Markt angehören.*

Der **dritte** Bereich erlaubt es, den angewandten Suchzeitraum des ProScreener festzulegen. Jeder Kurs kann nämlich nur in einem unterschiedlichen Zeitraum angezeigt werden, da jede ausgewählte Figur einer bestimmten Zeitskala nicht in einer anderen existieren kann.

Der **vierte** Bereich betrifft die Ergebnissortierung nach einem im Code definierten Kriterium (vgl. hierzu "[Ergebnissortierung und Filterkriterien](#)"). Mit wenig einschränkenden Bedingungen können die Ergebnisse eines ProScreener vielfältig sein (z.B. einige Tausende für den NASDAQ). In diesen Fällen ist es sinnvoll, die Ergebnisse nach einem ausgewählten Kriterium zu sortieren und anzuzeigen:

- ▶ die 30 höchsten Titel (des Sortierungskriteriums),
- ▶ die 30 niedrigsten Titel (des Sortierungskriteriums)

Beispiel: Nehmen wir an, unser Programm soll diejenigen Wertpapiere des SRD in täglicher Ansicht anzeigen, die ein Volumen über 20.000 haben. Die Anzahl der Suchergebnisse sollte logischerweise 30 Ergebnisse übersteigen. Um die Ergebnisse auf die 30 Titel zu beschränken, deren Variation am höchsten bzw. am niedrigsten ist, kann das Sortierungskriterium z.B. auf % der Variation festgelegt werden.

Der **fünfte** Bereich erlaubt es Ihnen, den Code in unserer [Strategien-Bibliothek](#) anderen Nutzern zur Verfügung zu stellen.





Sind diese 5 Parameter fertig definiert, klicken Sie zum Starten der Suche auf "Programm bestätigen". Wenn die Titel Ihren Bedingungen entsprechen, dann werden sie in einer Liste wie der folgenden angezeigt:

Suche ProScreener-Verwaltung

Anzahl an Ergebnissen (nur bis zu 30 werden angezeigt): 534

Name	Veränd %	Letzter	Volumen	↓ criteria
AIR FRANCE -KLM	-3,64%	9,530	207.007	0
ALTRAN TECHN.	-2,69%	2,53	159.575	0
BENETEAU	-2,12%	6,91	14.243	0
FIMALAC	-6,69%	27,91	1.311	0
CARBONE LORRAINE	-7,32%	17,84	57.590	0
NEXANS	-3,44%	41,58	30.942	0
CREDIT AGRICOLE	-8,22%	8,820	2.070k	0
SOPRA GROUP	-1,98%	23,26	3.001	0
ATOS ORIGIN	+0,94%	17,10	92.738	0
TELEPERFORMANCE	-0,94%	18,87	107.817	0
HERMES INTL	-1,48%	106,15	32.916	0
INFOGRAMES ENTERT.	-2,00%	4,90	27.061	0
METROPOLE TV	-1,83%	13,41	106.544	0
UBISOFT ENTERTAIN	-5,25%	16,43	194.193	0
TF1	-2,84%	10,96	187.621	0
RALLYE	-4,78%	14,74	70.137	0
RODRIGUEZ GROUP	-4,40%	1,74	83.220	0
BONDUELLE	-1,42%	59,61	1.262	0
FONC.FIN.PARTICIP.	-6,21%	24,62	1.002	0
AVENIR TELECOM	+0,00%	0,43	25.934	0
HAULOTTE GROUP	-5,53%	4,95	45.453	0
ALTEN	-3,59%	17,46	4.475	0
WAVECOM	-0,49%	8,08	6.227	0
IPSOS	-1,27%	19,42	3.165	0
JC DECAUX SA.	-3,63%	11,43	52.123	0
AIR LIQUIDE	-3,21%	61,350	283.370	0
RHODIA	-3,39%	4,28	211.447	0
CGG VERITAS	-5,49%	10,85	416.838	0
CARREFOUR	-2,77%	28,050	596.089	0
CNP ASSURANCES	-3,30%	52,77	39.405	0



## **Benutzung des Fensters 'Top Mover'**

Neben dem ProScreener stellt Ihnen ProRealTime zwei Listen mit Top Movern zur Verfügung. Die Top Mover sind Scan-Fenster in Echtzeit, deren Kriterien vor-definiert sind (im Gegensatz zum ProScreener, der gänzlich personalisierbar ist). Die beiden Top Mover-Fenster sind identisch und erlauben ein paralleles Suchen (z.B. Variation nach oben in einem Fenster und Variation nach unten im anderen).

Die in den Top Movern vorgeschlagenen Suchkriterien gehören 4 Kategorien an und erlauben folgende Informationen zu ermitteln:

- ▶ Die Abweichungen und Gaps.
- ▶ Die größten Variationen der Vorbörse.
- ▶ Die hauptsächlichsten Candlestick-Formationen.
- ▶ Die Spread und Volumen der Titel im Orderbuch.



Name	↑ Veränd % Gestern	Letzter	Volumen	Referenzkurs
PARIS RE	+1,30%	11,71	1	11,56
BOLLORE	+1,13%	80,90	1.175	80
BUREAU VERITAS	+0,41%	26,69	27.238	26,58
ATOS ORIGIN	+0,35%	17,00	97.558	16,94
GENERALE DE SANTE	+0,08%	12,00	6.595	11,99
VILMORIN & CIE	+0,05%	73,94	652	73,9

Um mehr über die Verwendung der Top Mover zu erfahren, empfehlen wir Ihnen sich das Video-Anleitung "[Top Mover Scanner: Entdeckt Preise und Umsatz der Top Movers oder Eröffnungsgaps](#)" anzusehen.



## Kapitel II: Die Programmierung des ProScreener

In diesem Kapitel werden die 5 ProBuilder-Befehle dargestellt, die dem ProScreener zugrunde liegen. Sie können diese Befehle auch in der Funktionen-Bibliothek im Bereich "ProScreener-Befehle" finden. Wir werden nacheinander vorstellen:

- ▶ Anweisung zum Suchen und Filtern.
- ▶ Volumen-Antizipation.
- ▶ Multi-Zeitraum-Suche.
- ▶ Multi-Titel-Suche.

### Anweisung zum Suchen und Filtern

Der Befehl "**SCREENER**" erlaubt es, die Suche zu starten. Er ist dem "**RETURN**" des ProBuilder gleichbedeutend und kann wie letzterer zusammen mit anderen, ihm folgenden Funktionen verwendet werden, die wir gemeinsam untersuchen werden.

"**SCREENER**" wird auf folgende Weise verwendet:

## **SCREENER**[Condition]

Greifen wir zu einem Beispiel:

Wir suchen alle Titel eines Marktes, deren Schlusspreis unbedingt unter dem niedrigen Bollinger Band liegt. Die Bollinger Bänder werden über 10 Perioden berechnet und auf den Schlusskurs angewendet.

```
c1 = (Close < BollingerDown[10](Close))  
SCREENER[c1]
```

Es ist möglich, die Suche um mehrere Bedingungen zu erweitern, die entweder gleichzeitig oder alternativ erfüllt werden. In diesem Fall würde die ProScreener-Syntax folgendermaßen aussehen:

**SCREENER**[Condition1 **AND** Condition2]

oder

**SCREENER**[Condition1 **OR** Condition2]



Beispiel:

Wir suchen Wertpapiere, deren Preis über dem Bollinger Kanal liegt und die einen Aufwärtstrend aufweisen.

```
REM Schlusspreis über oberem Bollinger Band
Condition1 = (Close > Bollingerup[20](Close))
REM Schlusspreis > Eröffnungspreis
Condition2 = Close > Open
REM MM7 > MM23
Condition3 = (Average[7](Close) > Average[23](Close))
SCREENER[Condition1 AND Condition2 AND Condition3]
```

Es ist weiterhin möglich, eine Konstante oder einen Indikator (vorprogrammiert oder selbst definiert) festzulegen, die sowohl als Filterkriterium als auch zur Sortierung der Ergebnisse dienen:

- ▶ Wenn die Suche mehr als 30 Ergebnisse ergibt, filtert der Indikator die anzuzeigenden Titel. In diesem Fall ermöglicht Ihnen der vierte Bereich des Programmierfensters, die Reihenfolge der Anzeige festzulegen (vgl. [Beschreibung Seite 5](#)).
- ▶ Wenn die Suche weniger als 30 Ergebnisse anzeigt, sortiert der Indikator die Ergebnisse auf der Basis dieses neuen Indikatoren.

Die zu benutzende Syntax für eine Konstante ist folgende:

**SCREENER**[c1](TypicalPrice)

Um einen vorprogrammierten Indikator zu benutzen, ist es notwendig, den Indikator zunächst einer Variablen hinzuzufügen (hier "Criteria" genannt):

**Criteria = RSI**[14](Close)

**SCREENER**[c1](Criteria)

Wenn man einen selbst programmierten Indikator heranzieht, dann muss die Anweisung "CALL", wie im [ProBuilder-Handbuch](#) beschrieben, benutzt werden.

**MyRSI = CALL "RSI" [14]**

**SCREENER**[c1](MyRSI)



Beispiel:

Ermitteln wir alle Titel eines Marktes, deren Volumen über 50.000 liegt. Wir programmieren einen Indikator, der es uns erlaubt abzuschätzen, ob der RSI in Überkauf- oder Überverkaufsperioden liegt. Wir zeigen die Ergebnisse des ProScreener auf der Grundlage dieses letztgenannten Indikatoren an.

```

c1 = Volume > 50000
IF RSI[20](Close) > 70 THEN
    Criteria = 1
ELSIF RSI[20](Close) < 30 THEN
    Criteria = -1
ENDIF
SCREENER[c1](Criteria AS "über Transaktionen")
    
```

Name	Veränd %	Volumen	über Transaktionen
BALDA AG O.N.	+0,00%	61.523	-1
BEIERSDORF AG O.N.	-2,12%	270.317	-1
BANK OF IRELAND	-2,11%	192.584	-1
HUGO BOSS AG VZO O.N.	-2,78%	115.072	-1
LEONI AG NA O.N.	-15,81%	199.848	-1
SGL CARBON AG O.N.	-4,55%	182.799	-1
QIAGEN NV EO -	-0,32%	168.091	-1
VIVAICON AG O.N.	-5,84%	105.413	-1
CRH PLC ORD	-4,34%	91.000	-1
HEIDELBERG.DRUCKMA.O.N	-0,50%	186.461	-1
HENKEL AG+CO.KGAA ST O	-3,29%	70.826	-1
HOCHTIEF AG	-3,95%	227.229	-1
ISHARES DAX (DE)	-3,73%	447.329	-1
ISHARES TECDAX (DE)	-2,22%	136.965	-1
ISH. DJ EURO STOXX 50	-4,35%	831.616	-1
MERCK KGAA O.N.	-0,02%	53.451	-1
READYMIX PLC	+0,00%	180.745	-1
RHEINMETALL AG	+0,46%	127.277	-1
DRILLISCH AG O.N.	-2,16%	67.168	-1
ISHARES PLC - S+P 500	-6,80%	338.965	-1
IS.IISHS DJ EO STX 5	-4,13%	182.423	-1
STADA ARZNEIMITT.VNA O	-3,27%	157.061	-1
SOFTWARE AG O.N.	-6,56%	61.954	-1

Es ist möglich, keine Klammern zu verwenden, wenn man den Befehl **"SORT BY"** einfügt:

**SCREENER[c1] SORT BY TypicalPrice**



Wenn wir dieser Spalte einen personalisierten Namen geben möchten, dann müssen wir die Funktion "AS" wie unten dargestellt verwenden:

**SCREENER[c1](TypicalPrice AS "typical price")**

### **Volumen-Antizipation**

Der Befehl "**EstimatedVolume**" erlaubt es, eine lineare Schätzung des Volumens des aktuellen Balkens zu erhalten. Genauer gesagt berechnen wir das geschätzte Volumen auf folgende Weise:

**geschätztes\_Volumen = Volumen \* Koeffizient\_Multiplikator**

oder

**Koeffizient\_Multiplikator = Anzeigedauer / Vergangene Zeit, während der sich der Candlestick zeichnet**

Dieser Befehl erweist sich also insbesondere dafür interessant, um geschätztes und reelles Volumen miteinander zu vergleichen.

Betrachten wir ein Beispiel:

Wenn man davon ausgeht, dass sich der letzte Candlestick in einer 10-Minuten-Ansicht seit 1 Minute zeichnet, dann wäre das geschätzte Volumen 10 mal so hoch wie das aktuelle Volumen.

Versuchen wir die Beziehung des antizipierten Volumens des Tages zum Volumen des Vortages zu berechnen (um repräsentative Ergebnisse zu erhalten, sollten Sie eine Intraday-Periode verwenden) :

**REM Schätzt das Volumen des Tages**

**Vol0 = EstimatedVolume**

**REM Findet das Volumen des Vortages**

**Vol1 = Volume[1]**

**REM Filterkriterium: Beziehung zwischen dem antizipierten Volumen des Tages und dem Volumen des Vortages**

**SCREENER (Vol0 / Vol1 AS "Volumen")**

### **Multi-Perioden-Suche**

Es ist möglich, eine Suche zu definieren, die die Kriterien in mehreren Zeiteinheiten überprüft. Dieses erlaubt die Überprüfung, ob eine Bedingung z.B. entweder kurz- oder mittelfristig erfüllt wird. Der entsprechende Befehl ist "**TIMEFRAME**" und seine Syntax sieht folgendermaßen aus:

**TIMEFRAME (Zeiteinheit der Periode)**



Die verschiedenen zur Verfügung stehenden Zeiteinheiten werden folgendermaßen in den Code übersetzt:

Bedeutung	Code
1 Minute	1 minute
2 Minuten	2 minutes
3 Minuten	3 minutes
5 Minuten	5 minutes
10 Minuten	10 minutes
15 Minuten	15 minutes
30 Minuten	30 minutes
1 Stunde	1 hour
2 Stunden	2 hours
3 Stunden	3 hours
4 Stunden	4 hours
Täglich	daily
Wöchentlich	weekly

Die der Linie "**TIMEFRAME**" folgenden Anweisungen werden nur in der festgelegten Zeitperiode gesucht. Es ist daher möglich, mehrere "**TIMEFRAME**"-Linien innerhalb desselben Codes einzufügen, um eine Multi-Perioden-Suche durchzuführen.

Greifen wir zu einem Beispiel:

Wir möchten alle Titel des CAC 40 ermitteln, die die unten genannten Bedingungen erfüllen:

- ▶ In wöchentlicher Anzeige hat der Indikator Williams %R bei 14 Perioden einen Wert zwischen 0 und -20
- ▶ In der 30-Minuten-Ansicht kreuzt ein steigender "Exponentieller Gleitender Durchschnitt" bei 20 Perioden jenen über 12 Perioden.

Wir können also Werte in wöchentlichen Überkauf-Bereichen entdecken, die aber kurzfristig einen Abwärtstrend aufweisen.

**TIMEFRAME(weekly)**

**Condition1 = Williams[14](Close) < 0 AND Williams[14](Close) > -20**

**TIMEFRAME(30 minutes)**

**Condition2 = ExponentialAverage[20](Close) CROSSES OVER ExponentialAverage[12](Close)**

**SCREENER[Condition1 AND Condition2]**



## Multi-Titel-Suche

Wir konnten feststellen, dass es beim ProScreener nötig ist, einen Markt (oder eine Liste) anzugeben, innerhalb dessen wir die Suche ausführen möchten.

Die Anweisung "**EQUITYFRAME**" erlaubt es, das Kriterium mit einem spezifischen, demselben Markt angehörenden, Titel in Beziehung zu setzen.

Sie wird auf folgende Weise benutzt :

### **EQUITYFRAME("Name des Marktes" , "ticker")**

Der Befehl ermöglicht es also Ergebnisse eines bestimmten Titels zu vergleichen, aber auch einen neuen Indikator zu erstellen, der als Filterbedingung verwendet wird. Zur Ausnutzung dieses Befehls:

- ▶ Darf ausschließlich auf Märkte angewandt werden, die in Echtzeit zur Verfügung stehen
- ▶ Kann nicht auf Indizes der hauptsächlichlichen Märkte angewandt werden
- ▶ Muss der spezifische Titel Teil desselben durchsuchten Marktes sein (Bereich 2 des ProScreener - Fensters)

Wenn Sie z.B. an 30 zufällig ausgewählten Titeln des CAC 40 den Schlusskurs des laufenden Balkens des AXA-Aktie (Tickersymbol "CS") anzeigen möchten, dann schreiben Sie:

```
EQUITYFRAME("Euronext Paris" , "CS")
```

```
MyClose = Close
```

```
SCREENER(MyClose)
```





Die verschiedenen Märkte entsprechen der in der unteren Tabelle aufgelisteten Syntax

<b>Code EQUITYFRAME</b>	<b>Name Börsenmarkt</b>
Aktien & ETFs AMEX	Aktien "Amex"
Aktien Australien	Aktien "Australien"
Aktien Spanien	Aktien "Spanien"
Aktien Amsterdam	Euronext Amsterdam
Aktien Belgien	Euronext Brüssel
Aktien Portugal	Euronext Lissabon
Aktien Frankreich	Euronext Paris
Forex 47 Paare	Forex
Aktien Italien	Aktien "Italien"
LSE	London Stock Exchange
Aktien NASDAQ	Aktien "Nasdaq"
Aktien NYSE	Aktien "Nyse"
Aktien Schweiz VirtX	Aktien "Schweiz"
Aktien Schweiz SWX	Aktien "Schweiz"
Aktien Deutschland	Aktien "Xetra"

Wie beim Befehl **"TIMEFRAME"** wird jeder Code, der auf **"EQUITYFRAME"** folgt, auf den zuletzt festgelegten Titel angewendet. Um zu den Daten des ausgewählten Marktes zurückzukehren, können wir im Bereich ProScreener schreiben:

### **EQUITYFRAME(default)**



Die beiden unten vorgestellten Funktionen sind komplementär und das folgende Beispiel verdeutlicht dies auf anschauliche Weise:

Sie möchten an 30 zufällig ausgewählten Titeln des CAC 40 den Indikator anzeigen, der den Unterschied vom Schlusskurs des laufenden Balkens der AXA-Aktie (Tickersymbol "CS") zum entsprechenden Schlusskurs der ausgewählten Wertpapiere aufweist.

```
REM Wir beginnen damit, die aus dem Titel des CAC 40 des Euronext Paris zu suchenden
Informationen zu ermitteln
EQUITYFRAME("Euronext Paris" , "CS")
MyClose = Close
REM Wir kommen zu den Titeln des ausgewählten Marktes in dem Menü "Listenauswahl" zurück
EQUITYFRAME(default)
REM Wir erhalten die gewünschten Informationen
CloseVal = MyClose - Close
REM Wir lassen uns das Ergebnis anzeigen
SCREENER(CloseVal AS "MyIndicator")
```

Beispiel: Dieser Screener erlaubt es uns, die Korrelation zwischen einem Titel und dessen Markt oder einem Index und dessen Markt graphisch darzustellen. Wir berechnen das Verhältnis zwischen zwei ausgewählten Titeln ("EQUITYFRAME"), sowie den Unterschied dieses Ratio im Verhältnis zum Vortag.

```
TIMEFRAME(Daily)
CloseVal = Close
EQUITYFRAME("NASDAQ" , "AMZN")
CloseInd = Close
EQUITYFRAME(default)
Ratio = (CloseVal / CloseInd) * 100
ForceRelative = (Ratio - Ratio[1]) * 100
SCREENER(ForceRelative AS "Force Relative")
```



## Kapitel III: Praktische Anwendungen

### Einfache Beispiele

#### 1) RSI 1 Stunde: Überverkauf

Der RSI ist ein Indikator in Bezug auf Überkauf/Überverkauf, der eine umgekehrte Trendbewegung ankündigen kann. Wir werden hier die Titel ermitteln, die sich im Überverkauf-Bereich befinden.

Der Überverkauf wird festgelegt durch: **RSI < 30**. Er ist umso deutlicher, je mehr sich der RSI Null nähert. Wir werden daher einen ProScreener erstellen, der alle Titel anzeigt, deren RSI unter 30 liegt.

Beispiel RSI im Überverkauf:

REM Berechnet den RSI bei 14 Balken

MyRSI = RSI[14]

REM Filter : RSI < 30

Filter = MyRSI < 30

SCREENER[Filter](MyRSI AS "RSI")

#### 2) RSI 1 Stunde: Überkauf

Wir suchen die Titel im Überkauf (festgelegt durch: **RSI > 70**). Er wird umso deutlicher, je stärker sich der RSI 100 nähert. Wir werden also einen Screener erstellen, der alle Titel anzeigt, deren RSI über 70 liegt.

Beispiel RSI im Überkauf:

REM Berechnet den RSI bei 14 Balken

MyRSI = RSI[14]

REM Filter : RSI > 70

Filter = MyRSI > 70

SCREENER[Filter](MyRSI AS "RSI")

#### 3) Kreuzen von steigenden Gleitenden Durchschnitten

Es handelt sich darum, eine Tendenz darzustellen, bei der ein kurzer Gleitender Durchschnitt einen langen Gleitenden Durchschnitt kreuzt.

Wir werden einen ProScreener erstellen, der alle Titel anzeigt, deren Gleitender Durchschnitt über 20 Tage denjenigen über 50 Tage steigend schneidet.

Wir werden weiterhin den Zeitpunkt des Unterschiedes zwischen diesen beiden Gleitenden Durchschnitten berechnen, um die Kraft des Kreuzungspunktes zu schätzen. Wenn diese Zahl Null nahe ist, dann ist die Überschneidung langsam und beide Durchschnitte sind fast parallel, was ein wenig signifikantes Signal ergibt. Dahingegen, je höher diese Zahl ist, desto "dynamischer" ist der Schneidepunkt. Ein typisches Beispiel ist das eines flachen Gleitenden Durchschnittes, der einen Gleitenden Durchschnitt mit einer deutlichen Aufwärtsbewegung schneidet.



Beispiel: Schneiden eines steigenden Durchschnitts über 50 Tage durch einen Durchschnitt über 20 Tage

REM Berechnung eines Gleitenden Durchschnitts über 20 Tage

**Sma20 = Average[20]**

REM Berechnung eines Gleitenden Durchschnitts über 50 Tage

**Sma50 = Average[50]**

REM Schätzt die relative Geschwindigkeit des kurzen Gleitenden Durchschnitts in Bezug auf den langen GD

**Geschwindigkeit = Momentum(Sma50 - Sma20) \* 100 / Close**

REM Wählt den Titel zum Zeitpunkt des Überschneidens

**Filter = Sma20 Crosses Over Sma50**

**SCREENER[Filter](Geschwindigkeit AS "Dynamisch")**

#### 4) Kreuzen von fallenden Gleitenden Durchschnitten

Wir werden einen ProScreener erstellen, der alle Titel anzeigt, deren Gleitender Durchschnitt über 20 Tage den Gleitenden Durchschnitt über 50 Tage fallend kreuzt.

Der Zeitpunkt wird auf dieselbe Weise festgelegt wie im vorigen Beispiel.

Beispiel: Kreuzen eines fallenden Durchschnitts über 50 Tage mit einem Durchschnitt über 20 Tage

REM Berechnung des Gleitenden Durchschnitts über 20 Tage

**Sma20 = Average[20]**

REM Berechnung des Gleitenden Durchschnitts über 50 Tage

**Sma50 = Average[50]**

REM Schätzt die relative Geschwindigkeit des kurzen Gleitenden Durchschnitts im Verhältnis zum langen GD

**Geschwindigkeit = Momentum(Sma20 - Sma50) \* 100 / Close**

REM Wählt den Titel zum Zeitpunkt des Überschneidens

**Filter = Sma20 Crosses Under Sma50**

**SCREENER[Filter](Geschwindigkeit AS "Dynamisch")**



 **Komplexere Beispiele****1) RSI und steigende Trendwende**

Wir stellen Ihnen hier ein System vor, das es erlaubt, in Echtzeit diejenigen Titel zu ermitteln, die darauf deuten, ihren Trend umzukehren und eine Bewegung großen Ausmaßes zu beginnen.

Traditionell interessieren sich Analysten für Indikatoren oder Figuren, die eine Überhitzung oder eine Trendwende ankündigen: Diese werden allerdings aus ihrem Kontext gelöst betrachtet (einfach, weil ihnen keine technischen Werkzeuge zur Verfügung standen, die dies besser machen würden).

Mit dem ProScreeener gibt es heutzutage keine technischen Grenzen mehr und es ist uns daher möglich, ein wahres auf dem RSI basierendes Ermittlungssystem zu erstellen. Der Ausgangspunkt ist einfach: wir suchen einen RSI im Überverkauf, dessen Trend sich in fallend umkehrt.

Wir werden also einen Filter erstellen, der diese zwei Bedingungen enthält und der sich in einer einzigen Zeile schreibt:

```
Filter = RSI < 30 AND Momentum[1](RSI) > 0
```

Jetzt haben wir die uns zunächst interessierenden Titel und werden diese sortieren, um die interessantesten Kapitalanlagen hervorzuheben. Hierzu werden diejenigen betrachtet, deren Fall im Vergleich zur üblichen Volatilität des Titels am deutlichsten war.

Beispiel: RSI und steigende Trendwende

```
REM Filtert die Titel, deren RSI im Überverkauf und Umkehrtrend ist
```

```
// Filterkriterium: RSI < 30 und steigend
```

```
Filter = RSI < 30 AND Momentum[1](RSI) > 0
```

```
REM Bestimmt die "Kraft" des fallenden Trends
```

```
REM Findet den höchsten Punkt über die 20 vorangegangenen Balken
```

```
Höhepunkt = Highest[20](High)
```

```
REM Bestimmt den Abfall seit diesem Punkt
```

```
Abfall = Höhepunkt - Close
```

```
REM Bestimmt die übliche Volatilität des Titels (True Range Medianwert über 3 Balken)
```

```
Norm = Summation[3](TR) - Highest[3](TR) - Lowest[3](TR)
```

```
Geschwindigkeit = Fall / Norm
```

```
REM Anzeige der Ergebnisse
```

```
SCREENER[Filter](Geschwindigkeit AS "fallende Kraft")
```



## 2) RSI und fallende Trendwende

Der Ausgangspunkt dieses ProScreener ist das Suchen eines RSI im Überkauf, der einen fallenden Umkehrtrend aufweist.

Wie zuvor erstellen wir einen Filter, der sich in der ProBuilder-Formelsprache folgendermaßen schreibt:

```
Filter = RSI > 70 AND Momentum[1](RSI) < 0
```

Auf gleiche Weise wie im vorherigen ProScreener werden diejenigen Titel ermittelt, deren Aufwärtsbewegung im Vergleich zur üblichen Volatilität am deutlichsten auffällt.

Beispiel: RSI und fallender Umkehrtrend

```
REM Filtert die Titel, deren RSI im Überkauf und Umkehrtrend ist
```

```
// Filterkriterium: RSI > 70 und abnehmend
```

```
Filter = RSI > 70 AND Momentum[1](RSI) < 0
```

```
REM Ermittelt die "Kraft" des Aufwärtstrends
```

```
// Findet den tiefsten Punkt über die 20 vorangegangenen Balken
```

```
Tiefstpunkt = Lowest[20](Low)
```

```
// Ermittelt den Fortschritt der Kurse von diesem Punkt an
```

```
Steigung = Close - Tiefstpunkt
```

```
// Ermittelt die übliche Volatilität des Titels (True Range-Medianwert über 3 Balken)
```

```
Norm = Summation[3](TR) - Highest[3](TR) - Lowest[3](TR)
```

```
Geschwindigkeit = Steigung / Norm
```

```
REM Topmover des Screening: die deutlichsten relativen Aufwärtsbewegungen
```

```
SCREENER[Filter](Geschwindigkeit AS "Aufwärtskraft")
```



### 3) Bullish engulfing

Bullish engulfing ist eine der bekanntesten Figuren des japanischen Candlesticks, benötigt allerdings offensichtlich die Einbeziehung des Kontextes (Existenz einer Umkehrtendenz).

Beginnen wir mit der Definition des Bullish engulfing:

- ▶ Der vorherige Candlestick ist schwarz.
- ▶ Der laufende Candlestick beginnt unterhalb des Körpers des vorangehenden Candlesticks.
- ▶ Der laufende Candlestick schließt oberhalb des vorangehenden Candlesticks.

Diese drei Bedingungen werden einfach auf folgende Weise geschrieben:

```
Filter = Close[1] < Open[1] AND Open < Close[1] AND Close > Open[1]
```

Um einen Abwärtstrend zu ermitteln, benutzen wir erneut den o.g. Code und grenzen den Zeithorizont ein wenig ein. Mit einem RSI 14 wäre es vernünftig, sich für das Fallen der Kurse über die vorangegangenen 20 Balken zu interessieren. In Bezug auf japanische Candlesticks zeigt die Erfahrung, dass 8 Balken ausreichend sind (diese äußerst reaktiven Strukturen sind theoretisch auf 7 Candlesticks begrenzt)

Beispiel: Bullish engulfing

```
REM Bestimmt die "Stärke" der Abwärtstendenz
```

```
// Findet den höchsten Punkt über die 8 vorangegangenen Balken
```

```
Höchstpunkt = Highest[8](High)
```

```
// Bestimmt den Abfall von diesem Punkt an
```

```
Abfall = Höchstpunkt - Close
```

```
// Bestimmt die übliche Volatilität des Titels (True Range-Medianwert über 3 Balken)
```

```
Norm = Summation[3](TR) - Highest[3](TR) - Lowest[3](TR)
```

```
REM Filterkriterium : bullish engulfing
```

```
Filter = Close[1] < Open[1] AND Open < Close[1] AND Close > Open[1]
```

```
Geschwindigkeit = Abfall / Norm
```

```
REM Top Mover des Screening: die auffälligsten relativen Abfälle
```

```
SCREENER[Filter](Geschwindigkeit AS "Kraft der Abwärtstendenz")
```



#### 4) Bear engulfing

Beginnen wir mit der Definition eines Bear engulfing:

- ▶ Vorangegangener Candlestick ist weiß.
- ▶ Laufender Candlestick öffnet oberhalb des Körpers des vorangegangenen Candlesticks.
- ▶ Laufender Candlestick schließt unterhalb des Körpers des vorangegangenen Candlesticks.

Diese drei Bedingungen werden auf folgende einfache Weise geschrieben:

```
Filter = Close[1] > Open[1] AND Open > Close[1] AND Close < Open[1]
```

In Bezug auf das Ermitteln des Aufwärtstrends werden wir den Code des RSI im fallenden Umkehrtrend wieder aufnehmen.

Beispiel: Bear engulfing

```
REM bestimmt die "Kraft" des Aufwärtstrends
```

```
// Findet den Tiefpunkt über die vorangegangenen 8 Balken
```

```
Tiefpunkt = Lowest[8](Low)
```

```
// Bestimmt die Entwicklung des Kurses von diesem Punkt an
```

```
Ansteigen = Close - Tiefpunkt
```

```
// Bestimmt die übliche Volatilität des Titels (True Range-Medianwert über 3 Balken)
```

```
Norm = Summation[3](TR) - Highest[3](TR) - Lowest[3](TR)
```

```
REM Filterkriterium: bear engulfing
```

```
Filter = Close[1] > Open[1] AND Open > Close[1] AND Close < Open[1]
```

```
Geschwindigkeit = Ansteigen / Norm
```

```
REM Top Mover des Screening: die auffälligsten relativen Steigerungen
```

```
SCREENER[Filter](Geschwindigkeit AS "Aufwärtskraft")
```





### 5) Bearish moving average crossing

Dieser ProScreener setzt sich aus 3 Bedingungen über mehrere Zeiteinheiten zusammen:

- ▶ **Kondition 1:** MACD in wöchentlicher Ansicht < 0.
- ▶ **Kondition 2:** MACD in steigender wöchentlicher Ansicht.
- ▶ **Kondition 3:** Stochastik in täglicher Ansicht < 30.

Es werden diejenigen Titel gesucht, die dem vom System vorgeschlagenen Kaufniveau STOP am nächsten sind (in diesem Fall: der höchste Kurs des Vortages).

Der ProScreener zeigt alle Titel an, die sich noch unter diesem Niveau befinden sowie jene, die es gerade überschritten haben und in der Nähe bleiben (nicht mehr als +5%).

Die angezeigten Ergebnisse beachten die Bedingungen in den vorgegebenen Zeiteinheiten (wöchentliche und tägliche Zeitskala).

Beispiel: Bearish moving average crossing

REM Kondition 1: MACD wöchentlich < 0 und steigend

**TIMEFRAME**(weekly)

**MyMACD** = **MACD**[12,26,9](Close)

**c1** = **MyMACD** < 0 **AND** **MyMACD** > **MyMACD**[1]

REM Kondition 2: Stochastik täglich < 30

**TIMEFRAME**(daily)

**MySTO** = **Stochastic**[14,3](Close)

**c2** = **MySTO** < 30

REM Niveau des Kaufstopps

**MyStop** = **High**[1]

REM Kriterium: Position des Kurses im Verhältnis zum STOP

**Criteria** = (Close / **MyStop** - 1) \* 100

REM Kondition 3: die Kurse unterhalb des STOP oder bei weniger als +5%

**c3** = **Criteria** < 5

**SCREENER**[c1 **AND** c2 **AND** c3](Criteria)



## 6) Bullish moving average crossing

Dieser ProScreener setzt sich aus 3 Bedingungen über mehrere Zeiteinheiten zusammen:

- ▶ **Kondition 1:** MACD ein wöchentlicher Zeitskala > 0.
- ▶ **Kondition 2:** MACD in abnehmender wöchentlicher Zeitskala.
- ▶ **Kondition 3:** Stochastik in täglicher Zeitskala > 70.

Es werden diejenigen Titel gesucht, die vom System vorgeschlagenen Verkaufsniveau STOP am nächsten sind. Dieses Niveau ist der tiefste Kurs des Vortages.

Der ProScreener zeigt alle Titel an, die sich noch oberhalb dieses Niveaus befinden und jene, die es gerade unterschritten haben und in der Nähe bleiben (nicht mehr als -5%).

Die Ergebnisse beachten die Bedingungen in den festgelegten Zeiteinheiten (wöchentliche und tägliche Zeitskala).

Beispiel: Bullish moving average crossing

REM Kondition 1: MACD wöchentlich > 0 und abnehmend

TIMEFRAME(weekly)

MyMACD = MACD[12,26,9](Close)

c1 = MyMACD > 0 AND MyMACD < MyMACD[1]

REM Kondition 2: Stochastik täglich > 70

TIMEFRAME(daily)

MySTO = Stochastic[14,3](Close)

c2 = MySTO > 70

REM Verkaufsniveau STOP

MyStop = Low[1]

REM Kriterium: Position des Kurses im Verhältnis zum STOP

Criteria = (Close / MyStop - 1) \* 100

REM Kondition 3: die Kurse oberhalb des STOP oder bei nicht mehr als -5%

c3 = Criteria > -5

SCREENER[c1 AND c2 AND c3](Criteria)



 **Index**

**A**

---

CODE	SYNTAX	FUNKTION
<b>Abs</b>	Abs(a)	Mathematische Funktion "Absoluter Wert"
<b>AccumDistr</b>	AccumDistr(close)	Klassische Kumulierung/Verlaufs-Indikator
<b>ADX</b>	ADX[N]	Indikator Average Directional Index oder "ADX" von n Perioden
<b>ADXR</b>	ADXR[N]	Indikator Average Directional Index Rate oder "ADXR" von n Perioden
<b>AND</b>	a AND b	Verknüpfung AND Operator
<b>AroonDown</b>	AroonDown[N]	Aroon Down Indikator von n Perioden
<b>AroonUp</b>	AroonUp[N]	Aroon Up Indikator von n Perioden
<b>Atan</b>	Atan(a)	Mathematische Funktion "Arctangent" von
<b>AS</b>	RETURN Result AS "ResultName"	Instruktion zur Benennung einer Linie oder Indikators im Chart. Verwendet mit "RETURN".
<b>Average</b>	Average[N](price)	Einfacher Gleitender Durchschnitt mit n Perioden
<b>AverageTrueRange</b>	AverageTrueRange[N](price)	"Average True Range" - True Range geglättet nach der Wilder Methode

**B**

---

CODE	SYNTAX	FUNKTION
<b>BarIndex</b>	BarIndex	zeigt an, wie viele Kerzen aller geladener Daten angezeigt werden
<b>BollingerBandWidth</b>	BollingerBandWidth[N](price)	Bollinger Bandbreite Indikator
<b>BollingerDown</b>	BollingerDown[N](price)	Unteres Bollingerband
<b>BollingerUp</b>	BollingerUp[N](price)	Oberes Bollingerband
<b>BREAK</b>	(FOR...DO...BREAK...NEXT) or (WHILE...DO...BREAK...WEND)	Instruktion bewirkt einen Abbruch der FOR oder WHILE Schleife



## C

CODE	SYNTAX	FUNKTION
<b>CALL</b>	myResult = CALL myFunction	Ruft einen benutzerdefinierten Indikator auf, um diesen im Programm zu verwenden
<b>CCI</b>	CCI[N](price) or CCI[N]	Commodity Channel Index Indikator
<b>ChaikinOsc</b>	ChaikinOsc[Ch1, Ch2](price)	Chaikin Oszillator
<b>Chandle</b>	Chandle[N](price)	Chande Momentum Oszillator
<b>ChandeKrollStopUp</b>	ChandeKrollStopUp[Pp, Qq, X]	Chande und Kroll Protection Stopp für Long Positionen
<b>ChandeKrollStopDown</b>	ChandeKrollStopDown[Pp, Qq, X]	Chande und Kroll Protection Stopp für Short Positionen
<b>Close</b>	Close[N]	Schlusskurs der aktuellen Kerze oder der n letzten Kerze
<b>COLOURED</b>	RETURN Result COLOURED(R,G,B)	Färbt einer eine Chartlinie mit einer selbst definierten Farbe nach RGB
<b>COS</b>	COS(a)	Kosinus-Funktion
<b>Crosses Over</b>	a Crosses Over b	Boolean Operator überprüft, ob eine Linie eine Andere von unten nach oben gekreuzt hat
<b>Crosses Under</b>	a Crosses Under b	Boolean Operator überprüft, ob eine Linie eine Andere von oben nach unten gekreuzt hat
<b>CUMSUM</b>	CUMSUM(price)	Errechnet einen bestimmten Kurs anhand der ganzen geladenen Daten
<b>CurrentDayOfWeek</b>	CurrentDayOfWeek	Abbildung des aktuellen Tages der Woche
<b>CurrentHour</b>	CurrentHour	Abbildung der aktuellen Stunde
<b>CurrentMinute</b>	CurrentMinute	Abbildung der aktuellen Minute
<b>CurrentMonth</b>	CurrentMonth	Abbildung des aktuellen Monats
<b>CurrentSecond</b>	CurrentSecond	Abbildung der aktuellen Sekunde
<b>CurrentTime</b>	CurrentTime	Abbildung der aktuellen Zeit (HHMMSS)
<b>CurrentYear</b>	CurrentYear	Abbildung des aktuellen Jahres
<b>CustomClose</b>	CustomClose[N]	Bedingung anpassbar im Einstellungsfenster des Charts (Standard: Schlusskurs)
<b>Cycle</b>	Cycle(price)	Kreis-Indikator



## D

CODE	SYNTAX	FUNKTION
<b>Date</b>	Date[N]	Gibt das Datum von jedem im Chart geladenen Balken aus
<b>Day</b>	Day[N]	Gibt den Tag von jedem im Chart geladenen Balken aus
<b>Days</b>	Days[N]	Zähler der Tage seit 1900
<b>DayOfWeek</b>	DayOfWeek[N]	Wochentag von jedem Balken
<b>Dclose</b>	Dclose(N)	Schlusskurs vom n-ten Tag vor dem heutigen Tag
<b>DEMA</b>	DEMA[N](price)	Doppelter exponentieller Gleitender Durchschnitt
<b>Dhigh</b>	Dhigh(N)	Hoch des n-ten Balkens vor dem aktuellen Balken
<b>DI</b>	DI[N](price)	Abbildung von DI+ minus DI-
<b>Dminus</b>	Dminus[N](price)	Abbildung des DI- Indikatoren
<b>Dplus</b>	Dplus[N](price)	Abbildung des DI+ Indikatoren
<b>Dlow</b>	Dlow(N)	Tiefstkurs des n-ten Tages vor dem heutigen Tag
<b>DO</b>	See FOR and WHILE	Optionaler Befehl in der FOR und WHILE Schleife zur Definition der Schleifenaktion
<b>Dopen</b>	Dopen(N)	Eröffnungskurs des n-ten Tages vor dem heutigen Tag
<b>DOWNTO</b>	See FOR	Befehl in der FOR Schleife, um in absteigender Reihenfolge die Schleife auszuführen
<b>DPO</b>	DPO[N](price)	Detrended Price Oszillator

## E

CODE	SYNTAX	FUNKTION
<b>EaseOfMovement</b>	EaseOfMovement[[]]	Ease of Movement Indikator
<b>ELSE</b>	See IF/THEN/ELSE/ENDIF	Befehl wird verwendet, um die zweite Bedingung der IF-bedingten Angabe aufzurufen
<b>ELSEIF</b>	See IF/THEN/ELSE/ENDIF	Steht für ELSE IF (wird innerhalb der bedingten Schleife verwendet)
<b>EMV</b>	EMV[N]	Ease of Movement Value Indikator
<b>EQUITYFRAME</b>	EQUITYFRAME ("Börsenplatz", "Ticker")	Sucht eine Bedingung in Bezug auf einen Titel desselben Börsenmarktes
<b>ENDIF</b>	See IF/THEN/ELSE/ENDIF	End-Befehl von IF-bedingter Angabe
<b>EndPointAverage</b>	EndPointAverage[N](price)	Endpunkt des Gleitenden Durchschnitts von einer



<b>EstimatedVolume</b>	EstimatedVolume	Anzeige eines zukünftig geschätzten Volumens
<b>Exp</b>	Exp(a)	Mathematische Funktion "Exponentiell"
<b>ExponentialAverage</b>	ExponentialAverage[N](price)	Exponentieller Gleitender Durchschnitt

## F-G

---

CODE	SYNTAX	FUNKTION
<b>FOR/TO/NEXT</b>	FOR i =a TO b DO a NEXT	FOR Schleife (berechnet alle Werte in an- (TO) oder absteigender Reihenfolge (DOWNTO))
<b>ForceIndex</b>	ForceIndex(price)	Force Index Indikator (ermittelt wer den Markt dominiert (Käufer oder Verkäufer))

## H

---

CODE	SYNTAX	FUNKTION
<b>High</b>	High[N]	Hoch des aktuellen Balkens oder des n-t letzten Balkens
<b>Highest</b>	Highest[N](price)	Höchster Kurs über eine vorher definierte Anzahl von Balken
<b>HistoricVolatility</b>	HistoricVolatility[N](price)	Historische Volatilität (oder statistische Volatilität)
<b>Hour</b>	Hour[N]	Abbildung der Stunde von jedem im Chart geladenen Balken

## I-J-K

---

CODE	SYNTAX	FUNKTION
<b>IF/THEN/ENDIF</b>	IF a THEN b ENDIF	Gruppe von bedingten Anweisungen ohne eine zweite Anweisung
<b>IF/THEN/ELSE/ENDIF</b>	IF a THEN b ELSE c ENDIF	Gruppe von bedingten Anweisungen
<b>IntradayBarIndex</b>	IntradayBarIndex[N]	Zählt, wie viele Balken an einem Tag angezeigt werden, anhand der ganzen geladenen Daten



## L

CODE	SYNTAX	FUNKTION
<b>LinearRegression</b>	LinearRegression[N](price)	Linearer Regressionsindikator
<b>LinearRegressionSlope</b>	LinearRegressionSlope[N](price)	Steigung des Linearen Regressionindikatoren
<b>Log</b>	Log(a)	Mathematische Funktion "Neperian Logarithmus"
<b>Low</b>	Low[N]	Tief des aktuellen Balkens oder von dem n-t letzten Balken
<b>Lowest</b>	Lowest[N](price)	Tiefstkurs über eine vorher definierte Anzahl von Balken

## M

CODE	SYNTAX	FUNKTION
<b>MACD</b>	MACD[S,L,Si](price)	Moving Average Convergence Divergence (MACD) als Diagrammform
<b>MACDline</b>	MACDLine[S,L](price)	MACD Linien Indikator
<b>MassIndex</b>	MassIndex[N]	Mass Index Indikator angewandt über N Balken
<b>Max</b>	Max(a,b)	Mathematische Funktion "Maximum"
<b>MedianPrice</b>	MedianPrice	Durchschnitt vom Hoch und vom Tief
<b>Min</b>	Min(a,b)	Mathematische Funktion "Minimum"
<b>Minute</b>	Minute	Abbildung der Minute von jedem im Chart geladenen Balken
<b>Mod</b>	a Mod b	Mathematische Funktion "Rest der Division"
<b>Momentum</b>	Momentum[N]	Momentum Indikator (Schluss – Schluss des n-t letzten Balken)
<b>MoneyFlow</b>	MoneyFlow[N](price)	MoneyFlow Indikator (Ergebnis zwischen -1 und 1)
<b>MoneyFlowIndex</b>	MoneyFlowIndex[N]	MoneyFlow Index Indikator
<b>Month</b>	Month[N]	Abbildung des Monats von jedem im Chart geladenen Balken



## N

---

CODE	SYNTAX	FUNKTION
<b>NEXT</b>	See FOR/TO/NEXT	End Befehl für die FOR Schleife
<b>NOT</b>	NOT a	Logischer Operator NOT

## O

---

CODE	SYNTAX	FUNKTION
<b>OBV</b>	OBV(price)	On-Balance-Volume-Indikator
<b>ONCE</b>	ONCE VariableName = VariableValue	Einführung einer Definitionsanweisung, welche nur einmal ausgeführt wird
<b>Open</b>	Open[N]	Eröffnung des aktuellen Balkens oder des n-t letzten Balkens
<b>OpenOfNextBar</b>	OpenOfNextBar[N]	Eröffnung des Balkens, welcher dem n-ten Balken folgt
<b>OR</b>	a OR b	Logischer Operator OR

## P-Q

---

CODE	SYNTAX	FUNKTION
<b>PriceOscillator</b>	PriceOscillator[S,L](price)	Prozentualer Preis Oszillator
<b>PositiveVolumeIndex</b>	PriceVolumeIndex(price)	Positiver Volumen Index Indikator
<b>PVT</b>	PVT(price)	Preis Volumen Trend Indikator





## R

CODE	SYNTAX	FUNKTION
<b>R2</b>	R2[N](price)	R-zum Quadrat Indikator (Fehlerquote der linearen Regression vom Kurs)
<b>Range</b>	Range[N]	berechnet die Spanne (Hoch minus Tief)
<b>REM</b>	REM comment	Einführung einer Bemerkung
<b>Repulse</b>	Repulse[N](price)	Repulse Indikator (misst die Stärke der Käufer und Verkäufer für jede Kerze)
<b>RETURN</b>	RETURN Result	Anweisung gibt das Ergebnis aus
<b>ROC</b>	ROC[N](price)	Preisrate des Change Indikator
<b>RSI</b>	RSI[N](price)	Relativer Stärke Index Indikator
<b>Round</b>	Round(a)	Mathematische Funktion "Rundet zur nächsten ganzen Zahl"

## S

CODE	SYNTAX	FUNKTION
<b>SAR</b>	SAR[At,St,Lim]	Parabolic SAR Indikator
<b>SARatdmf</b>	SARatdmf[At,St,Lim](price)	Geglätteter Parabolic SAR Indikator
<b>SCREENER</b>	SCREENER[c](Preis)	Anzeige des Ergebnisse
<b>Sin</b>	Sin(a)	Mathematische Funktion "Sinus"
<b>Sgn</b>	Sgn(a)	Mathematische Funktion "Zeichen von" einem (ist positiv oder negativ)
<b>SMI</b>	SMI[N,SS,DS](price)	Stochastik Momentum Index Indikator
<b>SmoothedStochastic</b>	SmoothedStochastic[N,K](price)	Geglätteter Stochastik
<b>SORT BY</b>	Screeener(c1) SORT BY price	Erlaubt die Anzeige oder das Filtern der Ergebnisse
<b>Square</b>	Square(a)	Mathematische Funktion "zum Quadrat"
<b>Sqrt</b>	Sqrt(a)	Mathematische Funktion "Quadratwurzel" von
<b>STD</b>	STD[N](price)	Statistische Funktion "Standardabweichung"
<b>STE</b>	STE[N](price)	Statistische Funktion "Standard Fehler"
<b>Stochastic</b>	Stochastic[N,K](price)	%K Linie des Stochastik Indikator
<b>Summation</b>	Summation[N](price)	Errechnet einen bestimmten Preis über der n-t letzten Kerze
<b>SuperTrend</b>	SuperTrend[STF,N]	Super Trend Indikator



## T

CODE	SYNTAX	FUNKTION
<b>Tan</b>	Tan(a)	Mathematische Funktion "Tangens" von
<b>TEMA</b>	TEMA[N](price)	Dreifacher exponentieller Gleitender Durchschnitt
<b>THEN</b>	See IF/THEN/ELSE/ENDIF	Anweisung folgt der ersten Bedingung von "IF"
<b>Time</b>	Time[N]	Abbildung der Zeit von jedem im Chart geladenen Balken
<b>TimeSeriesAverage</b>	TimeSeriesAverage[N](price)	Zeitliche Serie Gleitender Durchschnitt
<b>TIMEFRAME</b>	TIMEFRAME(P)	Erlaubt das Durchsuchen von anderen, nicht voreingestellten, Zeitperioden
<b>TO</b>	See FOR/TO/NEXT	Richtungsanweisung in der "FOR" Schleife
<b>Today</b>	Today[N]	Datum des n-Perioden Balkens vor dem aktuellen Balken
<b>TotalPrice</b>	TotalPrice[N]	(Schluss + Eröffnung + Hoch + Tief)/4
<b>TR</b>	TR(price)	True Range Indikator
<b>TriangularAverage</b>	TriangularAverage[N](price)	Dreieckiger Gleitender Durchschnitt
<b>TRIX</b>	TRIX[N](price)	Dreifach geglätteter exponentieller Gleitender Durchschnitt
<b>TypicalPrice</b>	TypicalPrice[N]	Abbildung des Durchschnittskurses (Durchschnitt vom Hoch, Tief und Schluss)

## U

CODE	SYNTAX	FUNKTION
<b>Undefined</b>	a = Undefined	Setzt den Wert einer Variablen auf unbestimmt

## V

CODE	SYNTAX	FUNKTION
<b>Variation</b>	Variation(price)	Differenz zwischen dem Schluss des letzten Balkens und dem Schluss des aktuellen Balkens in %
<b>Volatility</b>	Volatility[S, L]	Chaikin Volatilität
<b>Volume</b>	Volume[N]	Volumen Indikator
<b>VolumeOscillator</b>	VolumeOscillator[S,L]	Volume Oszillator
<b>VolumeROC</b>	VolumeROC[N]	Volumen des Price Rate Of Change



## W

---

CODE	SYNTAX	FUNKTION
<b>WeightedAverage</b>	WeightedAverage[N](price)	Abbildung des gewichteten Gleitenden Durchschnitts
<b>WeightedClose</b>	WeightedClose[N]	Durchschnittswert von (2*Schluss), (1*Hoch) und (1*Tief)
<b>WEND</b>	See WHILE/DO/WEND	End-Anweisung der WHILE Schleife
<b>WHILE/DO/WEND</b>	WHILE (condition) DO (action) WEND	WHILE Schleife
<b>WilderAverage</b>	WilderAverage[N](price)	Abbildung des Gleitenden Durchschnitts nach Wilder
<b>Williams</b>	Williams[N](close)	%R vom Williams Indikator
<b>WilliamsAccumDistr</b>	WilliamsAccumDistr(price)	Akkumulation/Verteilung des Williams Indikators

## X

---

CODE	SYNTAX	FUNKTION
<b>XOR</b>	a XOR b	Logischer Operator eXclusive OR

## Y

---

CODE	SYNTAX	FUNKTION
<b>Year</b>	Year[N]	Jahr des Balkens n Perioden vor dem aktuellen Balken
<b>Yesterday</b>	Yesterday[N]	Datum des Tages des vorangegangenen Balkens n Perioden vor dem aktuellen Balken

## Z

---

CODE	SYNTAX	FUNKTION
<b>ZigZag</b>	ZigZag[Zr](price)	Abbildung des Zig-Zag Indikator eingeführt in der Elliott Wellen Theorie
<b>ZigZagPoint</b>	ZigZagPoint[Zp](price)	Abbildung des Zig-Zag Indikator, der Elliott Wellen Theorie berechnet auf Zp Punkte



## Weitere

---

CODE	FUNKTION
+	Additions-Anweisung
-	Subtraktions-Anweisung
*	Multiplikations-Anweisung
/	Divisions-Anweisung
=	Istgleich-Anweisung
<>	Differenz-Anweisung
<	Strenge Unterlegenheits-Anweisung
>	Strenge Überlegenheits-Anweisung
<=	Unterlegenheits-Anweisung
>=	Überlegenheits-Anweisung
//	Leitet eine Kommentar-Zeile ein





**ProRealTime**.com  
Die führende Workstation für Trader